Lesson 05

We will now study how to use *Mathematica* in computations involving calculus.

The two pillars of calculus are derivative and integral. They are both derived from the notion of limit.

Limit

***) Sin[x]/x**
f[x_] := Sin[x]/x
f[0]

Table[{ 10^(-k), f[1. 10^(-k)] }, {k,0,5}]

Plot[f[x], {x, -10, 10}]

Limit[f[x], x -> 0]

***) Tan[x]**
Plot[Tan[x], {x, -5, 5}]
Limit[Tan[x], x -> Pi/2, Direction -> 1]
Limit[Tan[x], x -> Pi/2, Direction -> -1]

***) Sin[1/x]**
Plot[Sin[1/x], {x, -1, 1}]
Limit[Sin[1/x], x -> 0]

Derivative

Instantaneous rate of change of a function f(x).
Slope of the line tangent to the graph of f(x).

Limit[$\frac{f(x+h)-f(x)}{h}$, h-> 0]

If this limit does not exist, the function is not differentiable.

***) Clear[poly]**
poly[x_] := x^4 - x^3 + 2 x + 1
dpoly=D[poly[x], x]

Limit[(poly[x + h] - poly[x])/h, h -> 0]

What is the derivative of poly at x=0,1,2,3,4,5?
Table[{ x, dpoly }, {x,0,5}]

Clear[h]
h=0.01;

```
Table[ { x, (poly[ x+ h] - poly[x])/h}, {x,0,5}]
Clear[h]
h = 0.000001;
Table[ { x, (poly[ x + h] - poly[x])/h}, {x, 0, 5}]

D[poly[x],x]
D[poly[x], {x, 2}]
D[poly[x], {x, 3}]
poly'[x]
poly''[x]
poly'''[x]
```

***) Maximum and minimum**

```
Clear[f]
f[r_] :=  r/2 - Pi r^3;
Plot[f[r], {r, -0.5, 0.5}]

g = D[f[r], r];
Solve[ g == 0, r]
FindRoot[ g == 0, {r, 0.2}]

FindMaximum[f[r], {r, 0.2} ]
FindMaximum[f[r], {r, 0 } ]
FindMaximum[f[r], {r, -1 } ]

FindMinimum[f[r], {r, 0 } ]
```

Plot f in black and add a blue line passing through the maximum and a red line passing through the minimum

```
Plot[{f[r], f[1/Sqrt[6 Pi]],f[-1/Sqrt[6 Pi]]}, {r, -0.5, 0.5},PlotStyle -> {Black, Blue, Red}]
```

***) Partial derivative**

```
D[Exp[3 x^2 + y], x]
D[D[Sin[x y], x], y]
D[Sin[ x y], x, y]
```

Taylor Series

We want to find a polynomial P(x) that can approximate a function f(x) close to a point x=a. Polynomials are easy to work with and allows for analytical approximations.
We want:
P(a) = f(a)
P'(a)=f'(a)
P''(a) = f''(a)
...
but this cannot go on forever, because the higher order derivatives of the polynomial will eventually become zero.

Suppose:
P(x) = $a_0 + a_1(x-a) + a_2(x-a)^2 + \ldots + a_n(x-a)^n$

Since P(x) and f(x) share the same values of the derivatives:
$a_n = f^n$(a)/n!

The polynomial that approximates f(x) at x=a is the nth degree TAYLOR polynomial:
P(x) = $f(a) + f'(a)(x-a) + \frac{f''(a)}{2!}(x-a)^2 + \ldots + \frac{f(a)^{(n)}(a)}{n!}(x-a)^n$

(in the case of a=0, the Taylors series is also called Maclauring series)
Example:
Sin[x] close to x=0 is
Sin[x] ~ x - $\frac{x^3}{3!} + \frac{x^5}{5!}$ ...
It becomes easy to study the limit Sinx[x]/x

Make a plot of sin(x) and compare it with the first, third and fifth degree Taylor polynomial.

```
Clear[p1,p3,p5]
p1= Sum[ ( D[Sin[x] , {x, n} ] /. x -> 0 ) x^n/n!, {n, 0, 1}];
p3= Sum[ ( D[Sin[x] , {x, n} ] /. x -> 0 ) x^n/n!, {n, 0, 3}];
p5= Sum[ ( D[Sin[x] , {x, n} ] /. x -> 0 ) x^n/n!, {n, 0, 5}];

Plot[{Sin[x], p1,p3,p5}, {x, -4, 4}, PlotStyle -> {{Black, Thick}, Blue, Red, Green}]

Series[Sin[x], {x, 0, 7}]
Normal[Series[Sin[x], {x, 0, 7}]]
```

What is the limit of x/(Exp[x] - 1) when x approaches 0? Why?
*) Expand Exp[x] in a piece of paper to answer this question.
*) Double check it with: Series[Exp[x], {x, 0, 7}]
*) Then use Limit[x/(Exp[x] - 1) , x-> 0] to check your answer.

Integration

Defined via the Riemann sum, as the number of subintervals go to infinite and their individual lengths to zero.

*) **Indefinite integral**

```
Integrate[x, x]
Integrate[x^4 - 3 x^2, x]
```

*) **Definite integral**

```
Integrate[1/(1+x+x^2), {x, 0, 1}]
Integrate[Sin[x]^p, {x, 0, Pi}]
```

$\int_0^1 dx \int_0^{2-2x} (x^2 + y^2 + 1)\, dy =$

```
Integrate[Integrate[x^2 + y^2 + 1, {y, 0, 2-2x}],{x,0,1}]
Integrate[x^2 + y^2 + 1, {x,0,1}, {y, 0, 2-2x}]
```

*) **Assumptions**
Integrate[Sin[x]^p, {x, 0, Pi}, Assumptions -> Re[p] >-1]

∈ = \[Element]
Integrate[Sin[m x] Sin[n x], {x, 0, Pi}, Assumptions -> m ∈ Integers && n ∈ Integers && n==m]
Integrate[Sin[m x] Sin[n x], {x,0, Pi}, Assumptions →{ Element[n,Integers],Element[m,Integers], n == m}]
Assuming[{ m ∈ Integers, n ∈ Integers, n == m},Integrate[Sin[m x] Sin[n x], {x,0, Pi}]]

*) **NIntegrate: evaluates the integral using an adaptative algorithm, subdividing the interval of integration until a desired degree of accuracy is achieved.**
(Numerical integration)
Integrate[Exp[Sin[x]], {x, 0, 1}]
NIntegrate[Exp[Sin[x]], {x, 0, 1}]

Solve together in class problems (7iii, 7iv) and (8i,ii,iv)

*) **Integral as area: How to solve integrals without the command from *Mathematica*.**
(Riemann sums)

Integrate[x, {x, 4, 17}]
Plot[x, {x, 4, 17}, PlotRange -> {{0,20},{0,20}}, Filling → Axis]
Area = integral = (rectangle+triangle)= (17-4)*4 + (17-4)*(17-4)/2

Integrate[x^4 - 3 x^2, {x, 3, 8}]
Plot[x^4 - 3 x^2, {x, 3, 8}, PlotRange → {{0,10},{0,4500}},Filling → Axis]

Clear[bin];
bin = 1;
bin*((3 + bin/2.)^4 - 3 (3 + bin/2.)^2) + bin*((4 + bin/2.)^4 - 3 (4 + bin/2.)^2)+bin*((5 + bin/2.)^4 - 3 (5 + bin/2.)^2) + bin*((6 + bin/2.)^4 - 3 (6 + bin/2.)^2) + bin*((7 + bin/2.)^4 - 3 (7 + bin/2.)^2)
(* or equivalently *)
Clear[bin];
bin = 1;
Sum[bin ((x + bin/2.)^4 - 3 (x + bin/2.)^2) , {x, 3, 7}]
(* or equivalently *)
Clear[bin, f];
bin = 1;
f[x_] := x^4 - 3 x^2;
Sum[bin*f[x + bin/2.] , {x, 3, 7}]

*) more general:
Clear[bin,xin,xf];
bin = 0.5;
xin=3;

```
xf=8;
Sum[ bin ( (xin + bin (k - 1) + bin/2.)^4 - 3 (xin + bin (k - 1) + bin/2.)^2) , {k, 1, (xf - xin)/bin }]
Sum[ bin f[xin + bin (k - 1) + bin/2.] , {k, 1, (xf - xin)/bin }]

Clear[bin,xin,xf];
bin = 0.01;
xin=3;
xf=8;
Sum[ bin f[xin + bin (k - 1) + bin/2.] , {k, 1, (xf - xin)/bin }]

Clear[bin,xin,xf];
bin = 0.001;
xin=3;
xf=8;
Sum[ bin f[xin + bin (k - 1) + bin/2.] , {k, 1, (xf - xin)/bin }]
```

*) with DO-loop:

```
Clear[ bin, f ];
bin = 0.001;
f[ x_] := x^4 - 3 x^2;

Clear[xin,xf,integra];
xin=3;
xf=8;
integra=0.;
Do[
integra = integra + bin f[xin + (k-1) bin + bin/2];
,{k,1,(xf-xin)/bin}]
Print["The numerical integration results in ", integra]
```